Maple-assisted proof of formula for A209646

Robert Israel

7 March 2018

Of the $2^4 = 16$ possible configurations for a 1×4 sub-array, the following 9 avoid 0 0 1 and 1 0 0 horizontally.

```
> Rowconfigs:= select(A -> (A[1..3] <> [0,0,1]) and (A[1..3] <> [1,
  0,0]) and (A[2..4] <> [0,0,1]) and (A[2..4] <> [1,0,0]), [seq
  (convert(x,base,2)[1..4],x=2^4..2^5-1)]);
```

$$Rowconfigs := [[0, 0, 0, 0], [1, 0, 1, 0], [0, 1, 1, 0], [1, 1, 1, 0], [0, 1, 0, 1], [1, 1, 0, 1], [1, 0, 1, 1], [0, 1, 1, 1], [1, 1, 1, 1]] \quad (1)$$

Consider the $9^2 \times 9^2$ transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j), and 0 otherwise. The following Maple code computes it.

```
> Configs:= [seq(seq([Rowconfigs[i],Rowconfigs[j]],i=1..9),j=1..9)]
  :
  compat:= proc(i,j) local k,col;
     if Configs[i][2] <> Configs[j][1] then return 0 fi;
     for k from 1 to 4 do
        col:= [Configs[i][1][k],Configs[i][2][k],Configs[j][2][k]];
        if col = [0,0,1] or col = [1,0,1] then return 0 fi;
     od;
     1
  end proc:
  T:= Matrix(9^2,9^2,compat);
```

$$T := \left[\begin{array}{c} \textit{81 x 81 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{array}\right] \quad (2)$$

Thus for $n \geq 2$, $a(n) = u\, T^{n-2}\, v$ where u and v are 9^2-dimensional row and column vectors respectively of all 1's. The following Maple code produces these vectors.

```
> u:= Vector[row](9^2, 1):
  v:= Vector(9^2,1):
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^(n-2) . v,n=2..12);
```

$$81, 270, 630, 1215, 2079, 3276, 4860, 6885, 9405, 12474, 16146 \quad (3)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^5 - 4\,t^4 + 6\,t^3 - 4\,t^2 + t \quad (4)$$

This turns out to have degree 5 but no constant term. Thus we will have

$0 = u\,P(T)\ T^n\ v = \sum_{i=1}^{5} p_i\, a(i+n+2)$ for $n \geq 0$, where p_i is the coefficient of t^i in $P(t)$. This is a homogeneous linear recurrence of order 4.

```
> rec:= add(coeff(P(t), t, i)*a(i+n+2),i=1..5)=0;
```

$$rec := a(3+n) - 4\,a(4+n) + 6\,a(5+n) - 4\,a(6+n) + a(7+n) = 0 \quad \textbf{(5)}$$

The "empirical" formula $a(n) = 9\,n^3 + \frac{9}{2}\cdot n^2 - \frac{9}{2}\,n$ does satisfy this recursion:

```
> normal(eval(rec, a = (n -> 9*n^3 + 9/2*n^2 - 9/2*n)));
```

$$0 = 0 \quad \textbf{(6)}$$

Given that the formula is correct for the first four terms, it must also work for all the rest as well.